#include <ESP8266WiFi.h>      // 本程序使用 ESP8266WiFi库
#include <ESP8266WiFiMulti.h> //  ESP8266WiFiMulti库
#include <ESP8266WebServer.h> //  ESP8266WebServer库

#define buttonPin D3 // 按钮引脚D3

ESP8266WiFiMulti wifiMulti; // 建立ESP8266WiFiMulti对象,对象名称是'wifiMulti'

ESP8266WebServer esp8266_server(80); // 建立网络服务器对象，该对象用于响应HTTP请求。监听端口（80）

bool pinState; // 存储引脚状态用变量

void setup()
{
    Serial.begin(9600); // 启动串口通讯
    delay(10);
    Serial.println("");

    pinMode(buttonPin, INPUT_PULLUP); // 将按键引脚设置为输入上拉模式

    wifiMulti.addAP("301fang", "lyq1988w112233");
    Serial.println("Connecting ..."); // 则尝试使用此处存储的密码进行连接。
                                      // 另外这里只存储了3个WiFi信息，您可以存储更多
                                      // 的WiFi信息在此处。
    int i = 0;
    while (wifiMulti.run() != WL_CONNECTED)
    {                // 此处的wifiMulti.run()是重点。通过wifiMulti.run()，NodeMCU将会在当前
        delay(1000); // 环境中搜索addAP函数所存储的WiFi。如果搜到多个存储的WiFi那么NodeMCU
        Serial.print(i++);
        Serial.print(' '); // 将会连接信号最强的那一个WiFi信号。
    }                      // 一旦连接WiFI成功，wifiMulti.run()将会返回“WL_CONNECTED”。这也是
                           // 此处while循环判断是否跳出循环的条件。
    // WiFi连接成功后将通过串口监视器输出连接成功信息
    Serial.println('\n');           // WiFi连接成功后
    Serial.print("Connected to ");  // NodeMCU将通过串口监视器输出。
    Serial.println(WiFi.SSID());    // 连接的WiFI名称
    Serial.print("IP address:\t");  // 以及
    Serial.println(WiFi.localIP()); // NodeMCU的IP地址

    esp8266_server.begin();
    esp8266_server.on("/", handleRoot);
    esp8266_server.onNotFound(handleNotFound);

    Serial.println("HTTP esp8266_server started"); //  告知用户ESP8266网络服务功能已经启动
}

void loop()
{
    pinState = digitalRead(buttonPin); // 获取引脚状态
    esp8266_server.handleClient();     // 处理http服务器访问
}

/* 以下函数处理网站首页的访问请求。此函数为本示例程序重点1
详细讲解请参见太极创客网站《零基础入门学用物联网》
第3章-第2节“通过网络服务将开发板引脚状态显示在网页中”的说明讲解。*/
void handleRoot()
{ //处理网站目录“/”的访问请求
    esp8266_server.send(200, "text/html", sendHTML(pinState));
}

/*
建立用于发送给客户端浏览器的HTML代码。此代码将会每隔5秒刷新页面。
通过页面刷新，引脚的最新状态也会显示于页面中
*/
String sendHTML(bool buttonState)
{

    String htmlCode = "<!DOCTYPE html> <html>\n";
    htmlCode += "<head><meta http-equiv='refresh' content='5'/>\n";
    htmlCode += "<title>ESP8266 Butoon State</title>\n";
    htmlCode += "<style>html { font-family: Helvetica; display: inline-block; margin: 0px auto; text-align: center;}\n";
    htmlCode += "body{margin-top: 50px;} h1 {color: #444444;margin: 50px auto 30px;} h3 {color: #444444;margin-bottom: 50px;}\n";
    htmlCode += "</style>\n";
    htmlCode += "</head>\n";
    htmlCode += "<body>\n";
    htmlCode += "<h1>ESP8266 BUTTON STATE</h1>\n";

    if (buttonState)
    {
        htmlCode += "<p>Button Status: HIGH</p>\n";
    }
    else
    {
        htmlCode += "<p>Button Status: LOW</p>\n";
    }

    htmlCode += "</body>\n";
    htmlCode += "</html>\n";

    return htmlCode;
}

// 设置处理404情况的函数'handleNotFound'
void handleNotFound()
{                                                             // 当浏览器请求的网络资源无法在服务器找到时，
    esp8266_server.send(404, "text/plain", "404: Not found"); // NodeMCU将调用此函数。
}